
Component stories
    

Basic component documentation is in PhoenixStorybook.Story.

  
    
  
  Documentation


Component documentation is fetched from your component doc tags:
	For a live_component, fetches @moduledoc content.
	For a function component, fetches @doc content from the matching function.

If you are deploying phoenix_storybook in production with an Elixir release, make sure your
doc chunks are not stripped out from the release.
releases: [
  my_app_web: [
    strip_beams: [
      keep: ["Docs"]
    ]
  ]
]

  
    
  
  Variation groups


You may want to present different variations of a component in a single variation block.
It is possible using PhoenixStorybook.VariationGroup.

  
    
  
  Container


By default, each variation is rendered within a div in the storybook DOM.
You can pass additional HTML attributes or extend the class attribute.
# storybook/my_component.story.exs
defmodule Storybook.MyComponent do
  use PhoenixStorybook.Story, :component
  def container, do: {:div, class: "block"}
end
If you need further sandboxing you can opt in for iframe rendering.
	For function components, storybook will use the iframe srcdoc attribute (the whole iframe content
is inlined as an HTML attribute).
	For live components, storybook will use the typical iframe behavior, triggering an extra HTTP
request to fetch the iframe content.

# storybook/my_component.story.exs
defmodule Storybook.MyComponent do
  use PhoenixStorybook.Story, :component
  def container, do: :iframe
  # or def container, do: {:iframe, style: "display: inline; ..."}
end
ℹ️ Learn more on this topic in the sandboxing guide.

  
    
  
  Aliases & Imports


When using nested components or JS commands, you might need to reference other functions or
components. Whilst it is possible to use fully qualified module names, you might want to provide
custom aliases and imports.
Here is an example defining both:
defmodule NestedComponent do
  use PhoenixStorybook.Story, :component
  def function, do: &NestedComponent.nested_component/1

  def aliases, do: [MyStorybook.Helpers.JSHelpers]
  def imports, do: [{NestedComponent, nested: 1}]

  def variations do
    [
      %Variation{
        id: :default,
        slots: [
          """
          <.nested phx-click={JSHelpers.toggle()}>hello</.nested>
          <.nested phx-click={JSHelpers.toggle()}>world</.nested>
          """
        ]
      }
    ]
  end
end

  
    
  
  Templates


You may want to render your components within some wrapping markup. For instance, when your
component can only be used as a slot of another enclosing component.
Some components, such as modals, slideovers, and notifications, are not visible from the
start: they first need user interaction. Such components can be accompanied by an outer template,
that will for instance render a button next to the component, to toggle its visibility state.

  
    
  
  Variation templates


You can define a template in your component story by defining a template/0 function.
Every variation will be rendered within the defined template, the variation itself is injected
in place of <.psb-variation/>.
def template do
  """
  <div class="my-custom-wrapper">
    <.psb-variation/>
  </div>
  """
end
You can also override the template, per variation or variation_group by setting the :template key
to your variation. Setting it to a falsy value will disable templating for this variation.

  
    
  
  Variation group templates


Variation groups can also leverage on templating:
	either by wrapping every variation in their own template.

"""
<div class="one-wrapper-for-each-variation">
  <.psb-variation/>
</div>
"""
	or by wrapping all variations as a whole, in a single template.

"""
<div class="a-single-wrapper-for-all">
  <.psb-variation-group/>
</div>
"""
If you want to get unique id, you can use :variation_id that will be replaced, at rendering time
by the current variation (or variation group) id.

  
    
  
  Placeholder attributes


In template, you can pass some extra attributes to your variation. Just add them to the
.psb-variation or .psb-variation-group placeholder.
"""
<.form_for :let={f} for={%{}} as={:user}>
  <.psb-variation form={f}/>
</.form>
"""

  
    
  
  JS-controlled visibility


Here is an example of templated component managing its visibility client-side, by toggling CSS
classes through JS commands.
defmodule Storybook.Components.Modal do
  use PhoenixStorybook.Story, :component

  def function, do: &Components.Modal.modal/1

  def template do
    """
    <div>
      <button phx-click={Modal.show_modal()}>Open modal</button>
      <.psb-variation/>
    </div>
    """
  end

  def variations do
    [
      %Variation{
        id: :default_modal,
        slots: ["<:body>hello world</:body>"]
      }
    ]
  end
end

  
    
  
  Elixir-controlled visibility


Some components don't rely on JS commands but need external assigns, like a modal that takes a
show={true} or show={false} assign to manage its visibility state.
PhoenixStorybook handles special psb-assign and psb-toggle events that you
can leverage on to update some properties that will be passed to your components as extra assigns.
defmodule Storybook.Components.Slideover do
  use PhoenixStorybook.Story, :component
  def function, do: &Components.Slideover.slideover/1

  def template do
    """
    <div>
      <button phx-click={JS.push("psb-assign", value: %{show: true})}>
        Open slideover
      </button>
      <.psb-variation/>
    </div>
    """
  end

  def variations do
    [
      %Variation{
        id: :default_slideover,
        attributes: %{
          close_event: JS.push("psb-assign", value: %{variation_id: :default_slideover, show: false})
        },
        slots: ["<:body>Hello world</:body>"]
      }
    ]
  end
end

  
    
  
  Template code preview


By default, the code preview will render the variation and its template markup as well.
You can choose to render only the variation markup, without its surrounding template by using the
psb-code-hidden HTML attribute.
"""
<div psb-code-hidden>
  <button phx-click={Modal.show_modal()}>Open modal</button>
  <.psb-variation/>
</div>
"""

  
    
  
  Block, slots & let


Liveview let you define blocks of HEEx content in your components, referred to as as slots.
They can be passed in your variations with the :slots keys :
%Variation{
  id: :modal,
  slots: [
    """
    <:button>
      <button type="button">Cancel</button>
    </:button>
    """,
    """
    <:button>
      <button type="button">OK</button>
    </:button>
    """
  ]
}
You can also use LiveView let mechanism
to pass data to your default block. You just need to declare the let attribute you are using in
your variation.
%Variation{
  id: :list,
  attributes: %{stories: ~w(apple banana cherry)},
  let: :entry,
  slots: [
    "I like <%= entry %>"
  ]
}
let syntax can also be used with named slots, but requires no specific livebook setup.
%Variation{
  id: :table,
  attributes: %{
    rows: [
      %{first_name: "Jean", last_name: "Dupont"},
      %{first_name: "Sam", last_name: "Smith"}
    ]
  },
  slots: [
    """
    <:col :let={user} label="First name">
      <%= user.first_name %>
    </:col>
    """,
    """
    <:col :let={user} label="Last name">
      <%= user.last_name %>
    </:col>
    """
  ]
}

  
    
  
  Late evaluation


In some cases, you want to pass to your variation attributes a complex value which should be
evaluated at runtime but not in code preview (where you rather want to see the orignal expression).
For instance with the following variation of a Modal component.
%Variation{
  attributes: %{
    :"on-open": JS.push("open"),
    :"on-close": {:eval, ~s|JS.push("close")|}
  }
}
Both open & close events would work, but code would be rendered like this.
<.modal
  on-open="%Phoenix.LiveView.JS{ops: [["push", %{event: "open"}]]}"
  on-close={JS.push("close")}
/>

  
    
  
  Layout


You can control how your story variations are rendered in the stories tab by defining an optional layout/0 function in any of your component or live_component stories.
By default a story will be rendered with the :two_columns layout. But you can use the alternate :one_column layout making the component preview taking the full container width.
defmodule Storybook.Components.Breadcrumb do
  use PhoenixStorybook.Story, :component

  def layout, do: :one_column

  # ...
end


  

    
Custom Icons
    

You can provide custom sidebar & header icons for your stories.
Page stories can also provide iconized navigation tabs.
PhoenixStorybook gives you the ability to render icons with 2 different providers:
	FontAwesome which offers a decent set of free icons and a lot of
additional styles with paid plans
	HeroIcons wich offer hundreds of free high quality icons


  
    
  
  Declaring an icon


Whenever you provide an icon to the storybook, you should follow the following structure:
{:icon_provider, icon_name, :icon_style, additional_css_classes}.
The two last tuple elements are optional.
{:fa, "book"} # note that the FontAwesome icon name omits the fa- prefix
{:fa, "book", :solid} # same than previous one, :solid is the default style
{:fa, "skull", :duotone} # only for FontAwesome paid users
{:fa, "skull", :duotone, "psb-px-2"}
{:hero, "cake"} # for all HeroIcons
{:hero, "cake", :outline} # same than previous one, :outline is the default style
{:hero, "cake", :outline, "psb-w-2 psb-h-2"}

  
    
  
  FontAwesome icons


PhoenixStorybook is built with a very small subset of FontAwesome free icons. If you want to use
other FontAwesome icons you need to provide a valid FontAwesome kit id.
It can be either free or paid, so you also need to configure your FontAwesome plan.
# lib/my_app_web/storybook.ex
defmodule MyAppWeb.Storybook do
  use PhoenixStorybook,
    otp_app: :my_app,
    font_awesome_plan: :pro, # default value is :free
    font_awesome_kit_id: "foo8b41bar4625",
end

  
    
  
  HeroIcons


PhoenixStorybook delegates icon rendering to heroicons_elixir.
Make sure to add their dependency in your mix.exs file.
defp deps do
  [
    {:heroicons, "~> 0.5.0"}
  ]
end
You can now use whichever HeroIcon icon you want, based on the library function names.

  
    
  
  Custom CSS


The last tuple argument is an optional CSS class list you can pass to improve icon rendering.
Since the icon is rendered within the storybook layout, and not within your components sandbox, you
should use CSS classes supported by the storybook.
	any psb-w-* or psb-h-* class (TailwindCSS classes for height & width prefixed by psb-)
	any psb-p-* or psb-m-* class (padding & margin)
	any psb-text-color-###
	any fa-* (FontAwesome modifiers)



  

    
Sandboxing components
    

In PhoenixStorybook your components live within the storybook, so they share some context with
the storybook: styling and scripts.
While the original Storybook for React only relies on iframes,
we find them quite slow and don't want them to be the default choice.
This guide will explain:
	what JS context do your components share with the storybook?
	how is the storybook styled to prevent most styling clashes?
	how you should provide the style of your components with scoped styles.
	how to, as a last resort, enable iframe rendering.


  
    
  
  1. What JS context do your components share with the storybook?


PhoenixStorybook runs with Phoenix LiveView and therefore requires its LiveSocket. This
LiveSocket is the same used by your components: you just need to inject it with your own Hooks,
Params and Uploaders.
To do so, create a JS file that will declare your Hooks, Params and Uploaders and set them in
window.storybook. This script will be loaded immediately before the storybook's script.
:information_source: If you used mix phx.gen.storybook this file has already been created for you.

// assets/js/storybook.js
import * as Hooks from "./hooks";
import * as Params from "./params";
import * as Uploaders from "./uploaders";
(function () {
  window.storybook = { Hooks, Params, Uploaders };
})();
Then set the js_path: "/assets/storybook.js" option to the storybook within your storybook.ex
file. This is a remote path (not a local file-system path) which means this file should be served
by your own application endpoint with the given path.
You can also use this script to inject whatever content you want into document HEAD, such as
external scripts.
The Params will be available in page stories as connect_params assign.
There is currently no way to access them in component or live component stories.

  
    
  
  2. How is the storybook styled?


PhoenixStorybook is using TailwindCSS with
preflight (which means all default HTML styles from your
browser are removed) and a custom prefix:
psb- (which means that instead of using bg-blue-400 the storybook uses psb-bg-blue-400).
Only elements with the .psb class are preflighted, in order to let your component styling as-is.
So unless your components use psb or psb- prefixed classes there should be no styling leak from
the storybook to you components.

  
    
  
  3. How should you provide the style of your components?


You need to inject your component's stylesheets into the storybook. Set the
css_path: "/assets/storybook.css" option in storybook.ex. This is a remote path (not a local
file-system path) which means this file should be served by your own application endpoint with the
given path.
The previous part (2.) was about storybook styles not leaking into your components. This part is
about the opposite: don't accidentally mess up Storybook styling with your styles.
All containers rendering your components in the storybook (stories, playground, pages ...)
carry the .psb-sandbox CSS class and a custom sandboxing class of your choice.
You can leverage this to scope your styles with this class. Here is how you can do it with
TailwindCSS:
	configure phoenix_storybook with a custom sandbox_class:

# lib/my_app_web/storybook.ex
defmodule MyAppWeb.Storybook do
  use PhoenixStorybook,
    ...
    sandbox_class: "my-app-sandbox",
	use Tailwind important selector strategy
with this class. It will prefix all your tailwind classes increasing their specificity, hence
their priority.

// assets/tailwind.config.js
module.exports = {
  // ...
  important: ".my-app-sandbox",
};
	nest your custom styles under Tailwind @layer utilities. This way, your styling will also
benefit from sandboxing.

/* assets/css/storybook.css */
@tailwind base;
@tailwind components;
@tailwind utilities;

@layer utilities {
  /* this style will be generated as .my-app-sandbox * { ... } */
  * {
    font-family: "MyComponentsFont";
    @apply text-slate-600;
  }

  /* this style will be generated as .my-app-sandbox h1 { ... } */
  h1 {
    @apply text-2xl font-bold text-slate-700 mt-2 mb-6;
  }

  /* this style will be generated as .my-app-sandbox h2 { ... } */
  h2 {
    @apply text-xl font-bold text-slate-700 mt-2 mb-4;
  }
}

  
    
  
  4. Enabling iframe rendering


As a last resort, if for whatever reason you cannot make your component live within the storybook,
it is possible to enable iframe rendering, component per component.
This could be required e.g. if you need to bind listeners on document or when
you want to make sure responsive css works as expected.
Just add the iframe option to it.
# storybook/components/button.exs
defmodule MyAppWeb.Storybook.Components.Button do
 alias MyAppWeb.Components.Button
 use PhoenixStorybook.Story, :component

 def function, do: &Button.button/1
 def container, do: :iframe

 # ...
end


  

    
Manual setup
    

To start using PhoenixStorybook in your phoenix application you will need to follow these steps:
	Add the phoenix_storybook dependency
	Create your storybook backend module
	Add storybook access to your router
	Make your components assets available
	Create some content


  
    
  
  1. Add the phoenix_storybook dependency


Add the following to your mix.exs and run mix deps.get:
def deps do
  [
    {:phoenix_storybook, "~> 0.5.0"}
  ]
end

  
    
  
  2. Create your storybook backend module


Create a new module under your application lib folder:
# lib/my_app_web/storybook.ex
defmodule MyAppWeb.Storybook do
  use PhoenixStorybook,
    otp_app: :my_app,
    content_path: Path.expand("../storybook", __DIR__),
    # assets path are remote path, not local file-system paths
    css_path: "/assets/my_components.css",
    js_path: "/assets/my_components.js"
end

  
    
  
  3. Add storybook access to your router


Once installed, update your router's configuration to forward requests to a PhoenixStorybook
with a unique name of your choice:
# lib/my_app_web/router.ex
use MyAppWeb, :router
import PhoenixStorybook.Router
...
scope "/" do
  storybook_assets()
end

scope "/", PhoenixStorybookSampleWeb do
  pipe_through(:browser)
  ...
  live_storybook "/storybook", backend_module: MyAppWeb.Storybook
end

  
    
  
  4. Make your components assets available


Build a new CSS bundle dedicated to your live_view components: this bundle will be used both by your
app and the storybook.
In this README, we use assets/css/storybook.css as an example.
If your components require any hooks or custom uploaders, or if your pages require connect parameters, declare them as such in a new JS bundle:
// assets/js/storybook.js

import * as Hooks from "./hooks";
import * as Params from "./params";
import * as Uploaders from "./uploaders";

(function () {
  window.storybook = { Hooks, Params, Uploaders };
})();
These assets must be bundled and served by your own application. Our custom mix phx.gen.storybook
generator may guide you through these steps.
ℹ️ Learn more on this topic in the sandboxing guide.

  
    
  
  5. Create some content


Then you can start creating some content for your storybook. Storybook can contain different kinds
of stories:
	component stories: to document and showcase your components across different variations.
	pages: to publish some UI guidelines, framework or whatever with regular HTML content.
	examples: to show how your components can be used and mixed in real UI pages.

Stories are described as Elixir scripts (.story.exs) created under your :content_path folder.
Feel free to organize them in sub-folders, as the hierarchy will be respected in your storybook
sidebar.
Here is an example of a stateless (function) component story:
# storybook/components/button.story.exs
defmodule MyAppWeb.Storybook.Components.Button do
  alias MyAppWeb.Components.Button

  # :live_component or :page are also available
  use PhoenixStorybook.Story, :component

  def function, do: &Button.button/1

  def variations do [
    %Variation{
      id: :default,
      attributes: %{
        label: "A button"
      }
    },
    %Variation{
      id: :green_button,
      attributes: %{
        label: "Still a button",
        color: :green
      }
    }
  ]
  end
end


  

    
Visual Regression Testing
    

While we still encourage you writing regular unit test for your components, this doesn't protect
you against visual regressions.
Visual Regression Testing consists in taking automated screenshots of your components and compare
them pixel-per-pixel to notice any unwanted change.
For this we recommend using a dedicated tool such as percy.io.
This library provides you a dedicated enpoint to output your stories (only components' stories)
without the storybook main UI:
	single story endpoint: https://localhost:4000/storybook/visual_tests/buttons/button
	range story endpoint: https://localhost:4000/storybook/visual_tests?start=a&end=e

The last one renders all stories whose name starting between letter 'a' and letter 'e')


  

    
Theming components
    


  
    
  
  Theming Strategies


The storybook gives you different possibilities to apply a theme to your components. These
possibilities are named strategies.
The following strategies are available:
	sandbox class: set your theme as a CSS class, on the sandbox container, with a custom prefix
	assign: pass the theme as an assign to your components, with a custom key.
	function: call a custom module/function along with the current theme.

Here is how you can use these strategies. In your storybook.ex:
use PhoenixStorybook,
  themes_strategies: [
    sandbox_class: "prefix", # will set a class prefixed by `prefix-` on the sandbox container
    assign: :theme,
    function: {MyApp.ThemeHelper, :register_theme}
  ]
If the theme_strategies key is undefined, the default sandbox_class: "theme" strategy is applied.

  
    
  
  CSS theming


By default, the storybook is applying a theme-* CSS class to your components/page containers and
you should do as well to your application HTML body element.
It will allow you to style raw HTML elements
body.theme-colorful {
  font-family: // ...
}

.theme-colorful h1 {
  font-family: // ...
  font-size: // ...
}

  
    
  
  Using a Registry


This chapter explain how you can leverage on a Registry with the function theming strategy.
An effective way to store the current theme setting so that it can be available to all your
components, but still have different values for different (concurrent) users is to associate it to
the current LiveView pid.
Registry is a native Elixir module that handles decentralized storage, linked to specific
processes. We will leverage on this to associate a theme to the current LiveView pid.
First start a Registry from your Application module.
defmodule PhenixStorybook.Application do
  def start(_type, _args) do
    children = [
      {Registry, keys: :duplicate, name: ThemeRegistry}
    ]
  end
end
Then create a LiveView Hook that will fetch the theme from wherever it is relevant for your
application: database, user session, URL params... and store it in the Registry (it's working
because the Hook is running under the same pid than the Liveview).
defmodule ThemeHook do
  def on_mount(:default, params, _session, socket) do
    theme = current_user_theme(socket, params)
    Registry.register(ThemeRegistry, :theme, theme)
    {:cont, socket}
  end
end
Mount the hook in your router.
defmodule Router do
  live_session :default, on_mount: [ThemeHook] do
    scope "/" do
      # ...
    end
  end
end
Write a helper module, to be used from your components to fetch the current theme from the
Registry and merge it in the component's assigns.
defmodule ThemeHelpers do
  def set_theme(assigns) do
    pid_and_themes = Registry.lookup(ThemeRegistry, :theme)

    case find_by_pid(pid_and_themes, self()) do
      {_pid, theme} -> Map.put_new(assigns, :theme, theme)
      _ -> raise("theme not found in registry")
    end
  end

  defp find_by_pid(pid_and_themes, current_pid) do
    Enum.find(pid_and_themes, fn {pid, _} -> pid == current_pid end)
  end
end


  

    
PhoenixStorybook 
    



      
PhoenixStorybook provides a storybook-like UI interface for your
Phoenix components.
	Explore all your components, and showcase them with different variations.
	Browse your component's documentation, with their supported attributes.
	Learn how components behave by using an interactive playground.

[image: screenshot]
[image: screenshot]

  
    
  
  How does it work?


PhoenixStorybook is mounted in your application router and serves its UI at the mounting point of
your choice.
It performs automatic discovery of your storybook content under a specified folder (:content_path)
and then automatically generates a storybook navigation sidebar. Every module detected in your
content folder will be loaded and identified as a storybook entry.
Three kinds of stories are supported:
	component to describe your stateless function components or your live_components.
	page to write & document UI guidelines, or whatever content you want.
	example to show how your components can be used and mixed in real UI pages.


  
    
  
  Installation


To start using PhoenixStorybook in your phoenix application you will need to follow these steps:
	Add the phoenix_storybook dependency
	Run the generator


  
    
  
  1. Add the phoenix_storybook dependency


Add the following to your mix.exs and run mix deps.get:
def deps do
  [
    {:phoenix_storybook, "~> 0.6.0"}
  ]
end
[!IMPORTANT]
When picking a github version of the library (instead of an official hex.pm release) you
need the get the storybook's assets compiled.
To do so, please run mix dev.storybook.


  
    
  
  2. Run the generator


Run from the root of your application:
$> mix deps.get
$> mix phx.gen.storybook

And you are ready to go!
ℹ️ If you prefer manual setup, please read the setup guide.

  
    
  
  Configuration


Of all config settings, only the :otp_app, and :content_path keys are mandatory.
# lib/my_app_web/storybook.ex
defmodule MyAppWeb.Storybook do
  use PhoenixStorybook,
    # OTP name of your application.
    otp_app: :my_app,

    # Path to your storybook stories (required).
    content_path: Path.expand("../storybook", __DIR__),

    # Path to your JS asset, which will be loaded just before PhoenixStorybook's own
    # JS. It's mainly intended to define your LiveView Hooks in `window.storybook.Hooks`.
    # Remote path (not local file-system path) which means this file should be served
    # by your own application endpoint.
    js_path: "/assets/storybook.js",

    # Path to your components stylesheet.
    # Remote path (not local file-system path) which means this file should be served
    # by your own application endpoint.
    css_path: "/assets/storybook.css",

    # This CSS class will be put on storybook container elements where your own styles should
    # prevail. See the `guides/sandboxing.md` guide for more details.
    sandbox_class: "my-app-sandbox",

    # Custom storybook title. Default is "Live Storybook".
    title: "My Live Storybook",

    # Theme settings.
    # Each theme must have a name, and an optional dropdown_class.
    # When set, a dropdown is displayed in storybook header to let the user pick a theme.
    # The dropdown_class is used to render the theme in the dropdown and identify which current
    # theme is active.
    #
    # The chosen theme key will be passed as an assign to all components.
    # ex: <.component theme={:colorful}/>
    #
    # The chosen theme class will also be added to the `.psb-sandbox` container.
    # ex: <div class="psb-sandbox theme-colorful">...</div>
    #
    # If no theme has been selected or if no theme is present in the URL the first one is enabled.
    themes: [
      default: [name: "Default"],
      colorful: [name: "Colorful", dropdown_class: "text-pink-400"]

    # If you want to use custom FontAwesome icons.
    font_awesome_plan: :pro, # default value is :free
    font_awesome_kit_id: "foo8b41bar4625",
    font_awesome_rendering: :webfont, # default value is :svg

    # Story compilation mode, can be either `:eager` or `:lazy`.
    # It defaults to `:lazy` in dev environment, `:eager` in other environments.
    #   - When eager: all .story.exs & .index.exs files are compiled upfront.
    #   - When lazy: only .index.exs files are compiled upfront and .story.exs are compile when the
    #     matching story is loaded in UI.
    compilation_mode: :eager,
    
    # If you want to see debugging logs for storybooks compilation, set this to `true`. Default: `false`
    compilation_debug: true
  ]
All settings can be overridden from your config files.
# config/config.exs
config :my_app, MyAppWeb.Storybook,
  content_path: "overridden/content/path"
ℹ️ Learn more on theming components in the theming guide, on icons in the
icons guide.

Behaviour implemented by your backend module.

    PhoenixStorybook.Guides - phoenix_storybook v0.6.4
    
    

    


  
  

    
PhoenixStorybook.Guides 
    



      
This module is meant to be used from generated welcome.story.exs page.
It renders HTML markup from markdown guides located in the guides/folder.
Markup is precompiled because:
	we don't want to force user application to embed Earmark
	we don't want to put markdown guides in priv


  
    
  
  Examples


Guides.markup("components.md")
Guides.markup("icons.md")

PhoenixStorybook.Index 
    



      
An index is an optional file you can create in every folder of your storybook content tree to
improve rendering of the storybook sidebar.
The index files can be used:
	to customize the folder itself: name, icon and opening status.
	to customize folder direct children (only stories): name and icon.

Indexes must be created as index.exs files.
Read the icons guide for more information on custom icon usage.

  
    
  
  Usage


# storybook/_components.index.exs
defmodule MyAppWeb.Storybook.Components do
  use PhoenixStorybook.Index

  def folder_name, do: "My Components"
  def folder_icon, do: {:fa, "icon"}
  def folder_open?, do: true

  def entry("a_component"), do: [name: "My Component"]
  def entry("other_component"), do: [name: "Another Component", icon: {:fa, "icon", :thin}]
end

PhoenixStorybook.Rendering.CodeRenderer 
    



      
Responsible for rendering your components code snippet, for a given
PhoenixStorybook.Variation.
Uses the Makeup libray for syntax highlighting.

PhoenixStorybook.Rendering.ComponentRenderer 
    



      
Responsible for rendering your function & live components.

PhoenixStorybook.Rendering.RenderingContext 
    



      
A struct holding all data needed by ComponentRenderer and CodeRenderer to render story
variations.

PhoenixStorybook.Router 
    



      
Provides LiveView routing for storybook.

PhoenixStorybook.Stories.Attr 
    



      
An attr is one of your component attributes. Its structure mimics the LiveView 0.18.0 declarative
assigns.
Attributes declaration will populate the Playground tab of your storybook, for each of your
components.

PhoenixStorybook.Stories.Doc 
    



      
Functions to fetch component documentation and render it at HTML.

PhoenixStorybook.Stories.Slot 
    



      
A slot is one of your component slots. Its structure mimics the LiveView 0.18.0 declarative
assigns.
Slots declaration will populate the Playground tab of your storybook, for each of your
components.
Supported keys:
	id: the slot id (required). Should match your component slot name.
Use the id :inner_block for your component default slot.
	doc: a text documentation for this slot.
	required: true if the attribute is mandatory.

PhoenixStorybook.Stories.Variation 
    



      
A variation captures the rendered state of a UI component. Developers write multiple variations
per component that describe all the “interesting” states a component can support.
Each variation will be displayed in the storybook as a code snippet alongside with the
component preview.
Variations attributes type are checked against their matching attribute (if any) and will raise
a compilation an error in case of mismatch.
Advanced component & variation documentation is available in the
components guide.

  
    
  
  Usage


  def variations do
    [
      %Variation{
        id: :default,
        description: "Default dropdown",
        attributes: %{
          label: "A dropdown",
        },
        slots: [
          ~s|<:entry path="#" label="Account settings"/>|,
          ~s|<:entry path="#" label="Support"/>|,
          ~s|<:entry path="#" label="License"/>|
        ]
      }
    ]
  end

PhoenixStorybook.Stories.VariationGroup 
    



      
A variation group is a set of similar variations that will be rendered together in a single
preview <pre> block.

  
    
  
  Usage


  def variations do
    [
      %VariationGroup{
        id: :colors,
        description: "Different color buttons",
        variations: [
          %Variation{
            id: :blue_button,
            attributes: %{label: "A button", color: :blue }
          },
          %Variation{
            id: :red_button,
            attributes: %{label: "A button", color: :red }
          },
          %Variation{
            id: :green_button,
            attributes: %{label: "A button", color: :green }
          }
        ]
      }
    ]
  end

PhoenixStorybook.Story 
    



      
A story designates any kind of content in your storybook. For now only following kinds of stories
are supported :component, :live_component, and :page.
In order to populate your storybook, just create story scripts under your content path, and
implement their required behaviour.
Stories must be created as story.exs files.
In dev environment, stories are lazily compiled when reached from the UI.

  
    
  
  Usage



  
    
  
  Component


Implement your component as such.
Confer to:
	PhoenixStorybook.Variation documentation for variations.
	PhoenixStorybook.Attr documentation for attributes.

# storybook/my_component.exs
defmodule MyAppWeb.Storybook.MyComponent do
  use PhoenixStorybook.Story, :component

  # required
  def function, do: &MyAppWeb.MyComponent.my_component/1

  def attributes, do: []
  def slots, do: []
  def variations, do: []
end

  
    
  
  Live Component


Very similar to components, except that you need to define a component/0 function instead of
function/0.
# storybook/my_live_component.exs
defmodule MyAppWeb.Storybook.MyLiveComponent do
  use PhoenixStorybook.Story, :live_component

  # required
  def component, do: MyAppWeb.MyLiveComponent

  def attributes, do: []
  def slots, do: []
  def variations, do: []
end
ℹ️ Learn more on components in the components guide.

  
    
  
  Page


A page is a fairly simple story that can be used to write whatever content you want. We use it to
provide some UI guidelines.
You should implement the render function and an optional navigation function, if you want a tab
based sub-navigation. Current tab is passed as :tab in render/1 assigns.
# storybook/my_page.exs
defmodule MyAppWeb.Storybook.MyPage do
  use PhoenixStorybook.Story, :page

  def doc, do: "My page description"

  def navigation do
    [
      {:tab_one, "Tab One", {:fa, "book"}},
      {:tab_two, "Tab Two", {:fa, "cake", :solid}}
    ]
  end

  def render(assigns) do
    ~H"<div>Your HEEX template</div>"
  end
end

  
    
  
  Example


An example is a real-world UI showcasing how your components can be used and mixed in complex UI
interfaces.
Examples are rendered as a child LiveView, so you can implement mount/3, render/1 or any
handle_event/3 callback. Unfortunately handle_params/3 cannot be defined in a child LiveView.
By default, your example story's source code will be shown in a dedicated tab. But you can show
additional files source code by implementing the extra_sources/0 function which should return a
list of relative paths to your example related files.
# storybook/my_example.story.exs
defmodule MyAppWeb.Storybook.MyPage do
  use PhoenixStorybook.Story, :example

  def doc, do: "My page description"

  def extra_sources do
    [
      "./template.html.heex",
      "./my_page_html.ex"
    ]
  end

  def mount(_, _, socket), do: {:ok, socket}

  def render(assigns) do
    ~H"<div>Your HEEX template</div>"
  end
end

mix dev.storybook 
    



      
Make sure your storybook local dependency has all its assets packaged in priv.
$> mix dev.storybook

mix phx.gen.storybook 
    



      
Generates a Storybook and provides setup instructions.
$> mix phx.gen.storybook

The generated files will contain:
	a storybook backend in lib/my_app_web/storybook.ex
	a custom js file in assets/js/storybook.js
	a custom css file in assets/css/storybook.css
	scaffolding including example stories for your own storybook in storybook/

The generator supports the --no-tailwind flag if you want to skip the TailwindCSS specific bit.

